

**МИНОБРНАУКИ РОССИИ**  
**САНКТ-ПЕТЕРБУРГСКИЙ ГОСУДАРСТВЕННЫЙ**  
**ЭЛЕКТРОТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ**  
**«ЛЭТИ» ИМ. В.И. УЛЬЯНОВА (ЛЕНИНА)**  
**Кафедра МО ЭВМ**

**ОТЧЕТ**  
**по лабораторной работе №2**  
**по дисциплине «Операционные системы»**  
**Тема: Исследование интерфейсов программных модулей**

Студент гр. 8383

\_\_\_\_\_

Федоров И.А.

Преподаватель

\_\_\_\_\_

Ефремов М.А.

Санкт-Петербург

2020

### Цель работы.


Исследование интерфейса управляющей программы и загрузочных модулей. Исследование префикса сегмента программы (PSP) и среды, передаваемой программе.

### Ход работы.

Был написан и отлажен программный модуль типа **.COM** (см. рис. 1), который выбирает и распечатывает следующую информацию:

1. Сегментный адрес недоступной памяти, взятый из PSP в шестнадцатеричном виде.
2. Сегментный адрес среды, передаваемой программе.
3. Хвост командной строки.
4. Содержимое области среды.
5. Путь загружаемого модуля.

Результаты работы программы представлены на рис. 2 и 3. Исходный код **.COM** модуля приведен в приложении А.



```
DOSBox 0.74-3, Cpu speed: 3000 cycles, Frameskip 0, Program: DOSBOX

C:\MASM>masm LR_2COM.ASM
Microsoft (R) Macro Assembler Version 5.00
Copyright (C) Microsoft Corp 1981-1985, 1987. All rights reserved.

Object filename [LR_2COM.OBJ]:
Source listing [NUL.LST]:
Cross-reference [NUL.CRF]:

51518 + 448642 Bytes symbol space free

0 Warning Errors
0 Severe Errors

C:\MASM>link LR_2COM.OBJ

Microsoft (R) Overlay Linker Version 3.60
Copyright (C) Microsoft Corp 1983-1987. All rights reserved.

Run File [LR_2COM.EXE]:
List File [NUL.MAP]:
Libraries [.LIB]:
LINK : warning L4021: no stack segment

C:\MASM>_
```

Рисунок 1

```
DOSBox 0.74-3, Cpu speed: 3000 cycles, Frameskip 0, Program: DOSBOX
Microsoft (R) Overlay Linker Version 3.60
Copyright (C) Microsoft Corp 1983-1987. All rights reserved.

Run File [LR_2COM.EXE]:
List File [NUL.MAP]:
Libraries [L.LIB]:
LINK : warning L4021: no stack segment

C:\LETINICE\AVE_OC>EXE2BIN.EXE
Parse Error 2

C:\LETINICE\AVE_OC>EXE2BIN.EXE LR_2COM.EXE LR_2COM.COM

C:\LETINICE\AVE_OC>LR_2COM.COM

Address of close memory: 9FFF
Address of enviroment: 0188
Tail comand_line: (nothing)
Enviroment: PATH=Z:\
COMSPEC=Z:\COMMAND.COM
BLASTER=A220 I7 D1 H5 T6

Path: C:\LETINICE\AVE_OC\LR_2COM.COM
C:\LETINICE\AVE_OC>_
```

Рисунок 2 – Результат работы **.COM** модуля

```
DOSBox 0.74-3, Cpu speed: 3000 cycles, Frameskip 0, Program: DOSBOX

C:\LETINICE\AVE_OC>EXE2BIN.EXE LR_2COM.EXE LR_2COM.COM

C:\LETINICE\AVE_OC>LR_2COM.COM

Address of close memory: 9FFF
Address of enviroment: 0188
Tail comand_line: (nothing)
Enviroment: PATH=Z:\
COMSPEC=Z:\COMMAND.COM
BLASTER=A220 I7 D1 H5 T6

Path: C:\LETINICE\AVE_OC\LR_2COM.COM
C:\LETINICE\AVE_OC>LR_2COM.COM some_argument 123

Address of close memory: 9FFF
Address of enviroment: 0188
Tail comand_line: some_argument 123
Enviroment: PATH=Z:\
COMSPEC=Z:\COMMAND.COM
BLASTER=A220 I7 D1 H5 T6

Path: C:\LETINICE\AVE_OC\LR_2COM.COM
C:\LETINICE\AVE_OC>
```

Рисунок 3 – Результат работы с аргументами командной строки

## **Ответы на контрольные вопросы.**

### **Сегментный адрес недоступной памяти**

1) На какую область памяти указывает адрес недоступной памяти?

Сегментный адрес (равный 9FFFh) указывает на конец основной оперативной памяти DOS. Первые 640 Кбайт адресного пространства с сегментными адресами от 0000h до 9FFFh отведены под основную оперативную память (в которую может загружаться программа). За ней находится "верхняя память", зарезервированная DOS.

2) Где расположен этот адрес по отношению области памяти, отведенной программе?

Адрес расположен сразу за концом памяти, выделенной программе.

3) Можно ли в эту область памяти писать?

Эта область предназначена для размещения ПЗУ и других целей и зарезервирована, однако, т.к. в DOS нет механизма защиты памяти, то в нее можно писать и читать из нее.

### **Среда передаваемая программе**

1) Что такое среда?

Среда представляет собой текстовый массив, состоящий из строк вида:

"имя=параметр", 0

Здесь имя и параметр - текстовые величины, байт 0 завершает каждую строку. Имеется несколько стандартных переменных среды, например PATH (определяет пути к каталогам, в которых система ищет исполняемый файл), PROMPT (задает вид подсказки при диалоге с ОС).

2) Когда создается среда? Перед запуском приложения или в другое время?

Интерпретатор команд COMMAND.COM имеет свою среду, которую называют корневой средой (она создается при загрузке DOS). При запуске программы эта среда по умолчанию копируется для программы, однако может быть изменена в соответствии с требованиями для этой программы.

3) Откуда берётся информация, записываемая в среду?

Для этого используется пакетный файл AUTOEXEC.BAT, который устанавливает ключевые переменные среды (такие как PATH, PROMPT).

### **Выводы.**

В ходе работы были исследованы интерфейс управляющей программы и загрузочных модулей, префикс сегмента программы PSP, а так же среды, передаваемые программе.

# ПРИЛОЖЕНИЕ А

## КОД ДЛЯ СОМ ФАЙЛА

```

TESTPC      SEGMENT
             ASSUME CS:TESTPC, DS:TESTPC, ES:NOTHING, SS:NOTHING
             ORG 100H      ;обязательно!
START:      JMP MAIN

LEN_CLOSE_MEM EQU 30
LEN_ENV_SEG EQU 28
;TAIL db 83 DUP(?)
STR_CLOSE_MEM db 13,10, "Address of close memory:          $"
STR_ENV_SEG db 13,10, "Address of enviroment:              $"
STR_TAIL db 13,10, "Tail comand line: $"
STR_EMPTY_TAIL db " (nothing) $"
STR_ENVIROMENT_AREA db 13,10, "Enviroment: $"
STR_ENTER db 13,10, " $"
STR_PATH db 13,10, "Path: $"

;ПРОЦЕДУРЫ
;-----

WRITE_STR PROC near
    push ax
    mov ah, 09h
    int 21h
    pop ax
    ret
WRITE_STR ENDP
;-----

TETR_TO_HEX PROC near
    and     AL,0Fh
    cmp     AL,09
    jbe     NEXT
    add     AL,07
NEXT:      add     AL,30h
    ret
TETR_TO_HEX ENDP
;-----

BYTE_TO_HEX PROC near
; байт в AL переводится в два символа шестн. числа в AX
    push    CX
    mov     AH,AL
    call    TETR_TO_HEX
    xchg    AL,AH
    mov     CL,4
    shr     AL,CL
    call    TETR_TO_HEX ;в AL старшая цифра
    pop     CX          ;в AH младшая
    ret
BYTE_TO_HEX ENDP
;-----

WRD_TO_HEX PROC near
;перевод в 16 с/с 16-ти разрядного числа
; в AX - число, DI - адрес последнего символа
    push    BX
    mov     BH,AH
    call    BYTE_TO_HEX
    mov     [DI],AH
    dec     DI

```

```

        mov     [DI],AL
        dec     DI
        mov     AL,BH
        call    BYTE_TO_HEX
        mov     [DI],AH
        dec     DI
        mov     [DI],AL
        pop     BX
        ret
WRD_TO_HEX ENDP
;-----
BYTE_TO_DEC PROC near
; перевод в 10с/с, SI - адрес поля младшей цифры
        push    CX
        push    DX
        xor     AH,AH
        xor     DX,DX
        mov     CX,10
loop_bd: div     CX
        or      DL,30h
        mov     [SI],DL
        dec     si
        xor     DX,DX
        cmp     AX,10
        jae     loop_bd
        cmp     AL,00h
        je      end_l
        or      AL,30h
        mov     [SI],AL
end_l:   pop     DX
        pop     CX
        ret
BYTE_TO_DEC ENDP
;-----
;-----
; Funct lab_2

PRINT_ADDRESS_CLOSE_MEM PROC near
        push ax
        push dx
        mov ax, ds:[02h] ;в PSP
        mov di, offset STR_CLOSE_MEM
        add di, LEN_CLOSE_MEM
        call WRD_TO_HEX
        mov dx, offset STR_CLOSE_MEM
        call WRITE_STR
        pop dx
        pop ax
        ret
PRINT_ADDRESS_CLOSE_MEM ENDP

PRINT_ADDRESS_ENVIROMENT PROC near
        push ax
        push dx
        mov ax, ds:[2Ch] ; (44)
        mov di, offset STR_ENV_SEG
        add di, LEN_ENV_SEG
        call WRD_TO_HEX
        mov dx, offset STR_ENV_SEG
        call WRITE_STR
        pop dx

```

```

    pop ax
    ret
PRINT_ADDRESS_ENVIROMENT ENDP

```

```

PRINT_TAIL PROC near
    push ax
    push dx
    mov dx, offset STR_TAIL
    call WRITE_STR
    mov cx, 0
    mov cl, ds:[80h]          ;число СИМВОЛОВ В ХВОСТЕ
    cmp cl, 0
    je tail_empty
    mov di, 0
    xor dx,dx
print_tail_cycle:
    mov dl, ds:[81h+di]      ;ds+81h+di сделать потом
    mov ah,02H
    int 21h
    inc di
    loop print_tail_cycle
    jmp end_print
tail_empty:
    mov dx, offset STR_EMPTY_TAIL
    call WRITE_STR
end_print:
    pop dx
    pop ax
    ret
PRINT_TAIL ENDP

```

```

PRINT_PATH_ENVIROMENT PROC near
    push dx
    push ax
    push ds
    mov dx, offset STR_ENVIROMENT_AREA
    call WRITE_STR
    mov di, 0
    mov es, ds:[2Ch]
cycle_env:
    cmp byte ptr es:[di], 00h          ;mov dl, ds:[di] и сранивать dl с 0
    je enter_                          ;==
    mov dl, es:[di]
    mov ah, 02h
    int 21h
    inc di
    jmp cycle_env
enter_:
    inc di
    cmp word ptr es:[di], 0001h
    je path_
    mov dx, offset STR_ENTER
    call WRITE_STR
    jmp cycle_env
path_:
    inc di
    inc di
    mov DX, offset STR_PATH
    call WRITE_STR

```



```

cycle_p:
    cmp byte ptr es:[di], 00h
    je end_print_p
    mov dl, es:[di]
    mov ah, 02h
    int 21h
    inc di
    jmp cycle_p
end_print_p:
    pop dx
    pop ax
    pop ds
    ret
PRINT_PATH_ENVIROMENT ENDP

```

```

MAIN:
    call PRINT_ADDRESS_CLOSE_MEM
    call PRINT_ADDRESS_ENVIROMENT
    call PRINT_TAIL
    call PRINT_PATH_ENVIROMENT
    xor al, al
    mov AH, 4Ch
    int 21H
TESTPC ENDS
END START

```